

## ▼ Euklidische Vektorräume

Aufgaben: 6

```
> restart;  
with(LinearAlgebra):  
with(plots):
```

### ▼ Definitionen

**MATH:** Sei  $V$  ein  $\mathbb{R}$ -Vektorraum. Eine Abbildung  $\Psi: V \times V \rightarrow \mathbb{R}$  heißt (positiv definites) **Skalarprodukt**, falls gilt:

1.) (*bilinear*)  $\Psi(a \cdot v_1 + b \cdot v_2, w) = a \cdot \Psi(v_1, w) + b \cdot \Psi(v_2, w)$  für alle  $a, b \in \mathbb{R}$  und  $v_1, v_2, w \in V$  und

$\Psi(v, a \cdot w_1 + b \cdot w_2) = a \cdot \Psi(v, w_1) + b \cdot \Psi(v, w_2)$  für alle  $a, b \in \mathbb{R}$  und  $v, w_1, w_2 \in V$ .

2.) (*symmetrisch*)  $\Psi(v, w) = \Psi(w, v)$  für alle  $v, w \in V$ .

3.) (*positiv definit*)  $\Psi(v, v) > 0$  für alle  $v \in V$  mit  $v \neq 0$ .

Ist  $\Psi$  ein positiv definites Skalarprodukt auf dem endlich dimensionalen Vektorraum  $V$ , so heißt  $(V, \Psi)$  ein **Euklidischer Vektorraum** oder **endlich dimensionaler reeller Hilbertraum**. (Ist  $V$  nicht von endlicher Dimension, so spricht man von einem **Prähilbertraum**, einen **Hilbertraum** erhält man im unendlich-dimensionalen Fall durch Vervollständigung, ein Thema, das in der Analysis behandelt wird.)

**MATH:** Ein Vektor  $v \in V$  heißt **normiert**, falls  $\Psi(v, v) = 1$  gilt. Ist  $v \in V$  mit  $v \neq 0$  beliebig, so ist  $\frac{1}{\sqrt{\Psi(v, v)}} \cdot v$  normiert. Wir nennen diesen Vorgang **normieren**. Er

kann bei jedem Vektor  $v \neq 0$  in einem Euklidischen Vektorraum vorgenommen werden.

Wir können aber noch mehr: Wir können vermöge  $\Psi$  zwei normierte Vektoren  $v, w$  miteinander vergleichen. Man schreibt dann  $\Psi(v, w) =: \cos(\alpha)$  und nennt  $\alpha \in [0, \pi]$  den **Winkel** zwischen  $v$  und  $w$ . Dass dies möglich ist, folgt aus der **Cauchy-Schwarzschen Ungleichung**, die wir unten beweisen werden.

**MATH:** Wir nennen zwei Vektoren  $v, w$  in  $V$  **orthogonal** (zueinander) oder **senkrecht** (aufeinander) **bezüglich**  $\Psi$ , falls  $\Psi(v, w) = 0$  gilt. Eine Basis  $B \in V^n$  von  $V$  heißt **Orthogonalbasis**, falls je zwei verschiedene Elemente  $B_i$  und  $B_j$  aufeinander senkrecht stehen. Falls die Basisvektoren noch zusätzlich normiert sind, spricht man von einer **Orthonormalbasis** (kurz ONBasis).

**MATH:** Ist  $B \in V^n$  eine Basis von  $V$  und  $\Psi: V \times V \rightarrow \mathbb{R}$  eine Bilinearform, so heißt

die Matrix  $G \in \mathbb{R}^{n \times n}$  mit  $G_{i,j} = \Psi(B_i, B_j)$  die **Grammatrix** bezüglich  $B$ .

Im Falle einer linearen Abbildung haben wir die Abbildungsmatrix kennengelernt. Ein ähnlicher Fall liegt hier vor:

Die Grammatrix ermöglicht es uns, das Skalarprodukt zweier Vektoren  $v, w \in V$  durch ihre Koordinatenspalten und Matrixmultiplikationen auszurechnen:

$\Psi(v, w) = ({}^B v)^{tr} \cdot G \cdot {}^B w$ , falls  $G$  die Grammatrix bezüglich  $B$  ist.

Jeder Euklidische Vektorraum lässt eine Basis zu, bezüglich der die Grammatrix die Einheitsmatrix ist. Eine Basis ist genau von dieser Form, wenn sie eine ONBasis ist.

**BEISPIEL:**  $\mathbb{R}^4$  mit dem **Standardskalarprodukt** ist ein Euklidischer Vektorraum:

```
> Vector[row](4,symbol=a);  
Vector[row](4,symbol=b);
```

$$\begin{bmatrix} a_1 & a_2 & a_3 & a_4 \end{bmatrix}$$

$$\begin{bmatrix} b_1 & b_2 & b_3 & b_4 \end{bmatrix}$$

(1.1.1)

Das Standardskalarprodukt zweier Vektoren aus  $\mathbb{R}^4$  ist definiert durch:

```
> BilinearForm(Vector[row](4,symbol=a),Vector[row](4,symbol=b),conjugate=false);
```

$$a_1 b_1 + a_2 b_2 + a_3 b_3 + a_4 b_4$$

(1.1.2)

Wir betrachten die Standardbasis  $S$

```
> S:= [<1 | 0 | 0 | 0>, <0 | 1 | 0 | 0>, <0 | 0 | 1 | 0>, <0 | 0 | 0 | 1>];
```

$$S := \left[ \begin{bmatrix} 1 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 1 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 1 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 1 \end{bmatrix} \right]$$

(1.1.3)

und die zugehörige Grammatrix  ${}_S \Psi^S$ :

```
> Matrix(4,4,(i,j)-> BilinearForm(S[i],S[j],conjugate=false));
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

(1.1.4)

Wir betrachten eine andere Basis  $B$ :

```
> B:= [<1 | 1 | 1 | 1>, <1 | 1 | 1 | 2>, <1 | 1 | 0 | 3>, <1 | 0 | 0 | 4>];
```

$$B := \left[ \begin{bmatrix} 1 & 1 & 1 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 1 & 1 & 2 \end{bmatrix}, \begin{bmatrix} 1 & 1 & 0 & 3 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 & 4 \end{bmatrix} \right]$$

(1.1.5)

Die Grammatrix  ${}_B \Psi^B$  bezüglich dieser Basis:

```
> G:=Matrix(4,4,(i,j)-> BilinearForm(B[i],B[j],conjugate=false));
```

$$G := \begin{bmatrix} 4 & 5 & 5 & 5 \\ 5 & 7 & 8 & 9 \\ 5 & 8 & 11 & 13 \\ 5 & 9 & 13 & 17 \end{bmatrix} \quad (1.1.6)$$

**MATH:** Sind  $B$  und  $C$  Basen, und  ${}_B\Psi^B, {}_C\Psi^C$  die zugehörigen Grammatrizen. Dann gilt  ${}_B\Psi^B = ({}_C Id^B)^{tr} {}_C\Psi^C {}_C Id^B$ . Mit  ${}_B Id_C := ({}_C Id^B)^{tr}$  schreiben wir dann  ${}_B\Psi^B = {}_B Id_C {}_C\Psi^C {}_C Id^B$ .

### ÜBUNG [01]:

Es sei  $V := \mathbb{R}^2 \times 2$  mit  $\Psi: V \times V \rightarrow \mathbb{R}: (A, B) \mapsto \text{Spur}(A \cdot B^{tr})$  ein euklidischer Vektorraum.

1.) Bestimme eine Basis  $B$  von  $V$  und damit  ${}_B\Psi^B$ .

2.) Bestimme den Winkel zwischen den beiden folgenden Matrizen:

>  $A1 := \text{Matrix}([[1, 2], [3, 4]]);$   
 $A2 := \text{Matrix}([[7, 1], [5, -2]]);$

$$A1 := \begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}$$

$$A2 := \begin{bmatrix} 7 & 1 \\ 5 & -2 \end{bmatrix}$$

(1.1.7)

3.) Bestimme den Winkel zwischen den beiden folgenden Matrizen:

>  $B1 := \text{Matrix}([[1, 2], [2, 1]]);$   
 $B2 := \text{Matrix}([[0, -5], [5, 0]]);$

$$B1 := \begin{bmatrix} 1 & 2 \\ 2 & 1 \end{bmatrix}$$

$$B2 := \begin{bmatrix} 0 & -5 \\ 5 & 0 \end{bmatrix}$$

(1.1.8)

## ▼ Konstruktion einer Orthonormalbasis

**MATH:** Es gilt: Ist  $B \in V^n$  eine Basis von  $V$ , so gibt es eine ONBasis  $O$  von  $V$  mit  $\langle B_1, \dots, B_k \rangle = \langle O_1, \dots, O_k \rangle$

für  $k = 1, \dots, n$ .

Bei der Konstruktion dieser ONBasis hat man nur wenige Möglichkeiten:  $O_1$  ist die Normierung von  $B_1$  oder das Negative hiervon, wir entscheiden uns für

$$O_1 = \frac{1}{\sqrt{\Psi(B_1, B_1)}} B_1$$

Um  $O_2$  zu bekommen, setzen wir

$$O'_2 = a \cdot O_1 + B_2$$

und bestimmen  $a$  so, dass  $O'_2$  senkrecht auf  $O_1$  steht:

$$0 = \Psi(O_1, O'_2) = a \cdot \Psi(O_1, O_1) + \Psi(O_1, B_2) = a + \Psi(O_1, B_2).$$

Wir setzen also  $a = -\Psi(O_1, B_2)$  und bekommen Orthogonalität. Anschließend normieren wir  $O'_2$  und erhalten  $O_2$ .

Um  $O_3$  zu bekommen, setzen wir

$$O'_3 = a \cdot O_1 + b \cdot O_2 + B_3$$

und bestimmen auf dieselbe Weise  $a$  und  $b$  so, dass  $O'_3$  senkrecht auf  $O_1$  und  $O_2$  steht. Anschließend normieren wir wieder, um  $O_3$  zu bekommen. Dann fahren wir auf dieselbe Weise mit  $O_4, \dots, O_n$  und haben eine ONBasis bekommen.

Dieses Verfahren heißt auch **Gram-Schmidtsches Orthogonalisierungsverfahren** oder **Gram-Schmidtsches Orthonormalisierungsverfahren**.

**DENKANSTOSS:** Es sind genau  $2^n$  Wahlen für die ONBasis mit der obigen Eigenschaft möglich.

### ÜBUNG [02]:

Betrachte den  $\mathbb{R}^4$  mit dem Standardskalarprodukt und folgender Basis

$$\mathbf{B} := \left[ \langle 1 \mid 1 \mid 1 \mid 1 \rangle, \langle 1 \mid 1 \mid 1 \mid 0 \rangle, \langle 1 \mid 1 \mid 0 \mid 0 \rangle, \langle 1 \mid 0 \mid 0 \mid 0 \rangle \right];$$

$$B := \left[ \begin{bmatrix} 1 & 1 & 1 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 1 & 1 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 1 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 & 0 \end{bmatrix} \right] \quad (1.2.1)$$

1) Bestimme die Grammatrix  $\mathbf{G}$  bezüglich  $\mathbf{B}$ .

2) Bestimme nach dem obigen Verfahren eine ON-Basis  $\mathbf{C}$  von der Basis  $\mathbf{B}$ .

3) Wie muss man eine Basiswechselmatrix zwischen den Basen  $\mathbf{B}$  und  $\mathbf{C}$  anwenden, damit man die zu einer ON-Basis passende Einheitsmatrix erhält?

**MATH:** Man sieht, dass man aus dem Gram-Schmidtschen Verfahren auch dann eine Orthonormalbasis bekommt, wenn man nur mit einem endlichen Erzeugendensystem  $B$  (und nicht unbedingt mit einer Basis) anfängt.

## Winkel, Orthogonalprojektionen, Approximationen

**MATH:** Als Folgerung bekommen wir die **Cauchy-Schwarzsche Ungleichung:**

$$\Psi(v, v) \cdot \Psi(w, w) - \Psi(v, w)^2 \geq 0$$

mit Gleichheit genau dann, wenn  $v$  und  $w$  linear abhängig sind.

**Beweis:**

Wir können von einer ONBasis  $(B_1, B_2)$  von  $\langle v, w \rangle$  ausgehen, so dass

$$v = a \cdot B_1$$

$$w = b \cdot B_1 + c \cdot B_2.$$

Die linke Seite ist dann

$$a^2 \cdot (b^2 + c^2) - (a \cdot b)^2 = a^2 \cdot c^2$$


und die Behauptung folgt.

Insbesondere ist der Winkel zwischen zwei Vektoren  $v, w$ , die beide von Null verschieden sind, wohldefiniert.

Wir betrachten den Fall zweier Einheitsvektoren  $v = (1, 0)$  und

$$w = \left( \cos\left(\frac{\pi}{3}\right), \sin\left(\frac{\pi}{3}\right) \right)$$

```
> d1:=plot([cos(t),sin(t),t=0..2*Pi]:  
d2:=plot([cos(Pi/3),t*sin(Pi/3),t=0..1]):  
d3:=plot([t*cos(Pi/3),t*sin(Pi/3),t=0..1]):  
display([d1,d2,d3], scaling=constrained);
```



Wir sehen: Die **Orthogonalprojektion** von  $w$  auf  $\langle v \rangle$  (entlang dem zu  $\langle v \rangle$  senkrechten Teilraum) ist  $\Psi(v, w) \cdot v$ .

> **BilinearForm**( $\langle 1, 0 \rangle, \langle \cos(\text{Pi}/3), \sin(\text{Pi}/3) \rangle, \text{conjugate=false}$ )\*  
 $\langle 1, 0 \rangle$ ;

$$\begin{bmatrix} \frac{1}{2} \\ 0 \end{bmatrix}$$

**(1.3.1)**

Zur genauen Definition der Orthogonalprojektion siehe unten.

**MATH:** Mit dieser geometrischen Interpretation des Skalarproduktes können wir das Gram-Schmidtsche Verfahren geometrisch sehr leicht verstehen: Durch Subtraktion der Orthogonalprojektion von  $B_2$  auf  $O_1$  erhalten wir einen Vektor, der senkrecht auf  $O_1$  steht, etc.

**BEISPIEL:** Welche Polynomfunktion  $f$  vom Grad  $\leq 3$  der Form  
 $x \mapsto x^3 + a \cdot x^2 + b \cdot x + c$  liefert einen minimalen Wert für

$$\sum_{i=0}^4 f(i)^2?$$

Die Basis

```
> B:=[1,x,x^2];
```

$$B := [1, x, x^2] \quad (1.3.2)$$

des Teilraumes der Polynomfunktionen vom Grad  $< 3$  wird zuerst in eine ON-Basis verwandelt: Um die Quadratsummen ausdrücken zu können nehmen wir als Skalarprodukt

```
> SkPr:=proc(f::polynom,g::polynom)
>   return add(subs(x=i,f*g),i=0..4);
> end proc;
```

und benutzen das Gram-Schmidtsche Orthogonalisierungsverfahren:

```
> a:='a':b:='b':c:='c':
```

```
> o[1]:=B[1];
```

$$o_1 := 1 \quad (1.3.3)$$

```
> o[1]:=simplify(1/sqrt(SkPr(o[1],o[1]))*o[1]);
```

$$o_1 := \frac{1}{5} \sqrt{5} \quad (1.3.4)$$

```
> o[2]:=a*o[1]+B[2];
```

$$o_2 := \frac{1}{5} a \sqrt{5} + x \quad (1.3.5)$$

```
> o[2]:=subs(isolate(SkPr(o[1],o[2]),a),o[2]);
```

$$o_2 := -2 + x \quad (1.3.6)$$

```
> o[2]:=simplify(1/sqrt(SkPr(o[2],o[2]))*o[2]);
```

$$o_2 := \frac{1}{10} \sqrt{10} (-2 + x) \quad (1.3.7)$$

```
> o[3]:=a*o[1]+b*o[2]+B[3];
```

$$o_3 := \frac{1}{5} a \sqrt{5} + \frac{1}{10} b \sqrt{10} (-2 + x) + x^2 \quad (1.3.8)$$

```
> o[3]:=subs(isolate(SkPr(o[1],o[3]),a),o[3]);
```

$$o_3 := -6 + \frac{1}{10} b \sqrt{10} (-2 + x) + x^2 \quad (1.3.9)$$

```
> o[3]:=subs(isolate(SkPr(o[2],o[3]),b),o[3]);
```

$$o_3 := x^2 - 4x + 2 \quad (1.3.10)$$

```
> o[3]:=simplify(1/sqrt(SkPr(o[3],o[3]))*o[3]);
```

$$o_3 := \frac{1}{14} \sqrt{14} (x^2 - 4x + 2) \quad (1.3.11)$$

Der Teilraum  $\langle 1, x, x^2 \rangle$  enthält eine Lösung des Problems, nämlich 0. Wir suchen das Element der Form  $n := x^3 + a \cdot x^2 + b \cdot x + c$ , das im Sinne unseres Skalarprodukts "am nächsten" an diesem Teilraum liegt. Dazu suchen wir das eindeutige  $n$ , das senkrecht auf diesem Teilraum steht:

$$\begin{aligned} > n:=a*o[1]+b*o[2]+c*o[3]+x^3; \\ n:=x^3 + \frac{1}{5} a\sqrt{5} + \frac{1}{10} b\sqrt{10} (-2+x) + \frac{1}{14} c\sqrt{14} (x^2-4x+2) \end{aligned} \quad (1.3.12)$$

$$\begin{aligned} > n:=subs(isolate(SkPr(o[1],n),a),n); \\ n:=x^3 - 20 + \frac{1}{10} b\sqrt{10} (-2+x) + \frac{1}{14} c\sqrt{14} (x^2-4x+2) \end{aligned} \quad (1.3.13)$$

$$\begin{aligned} > n:=subs(isolate(SkPr(o[2],n),b),n); \\ n:=x^3 + \frac{54}{5} - \frac{77}{5} x + \frac{1}{14} c\sqrt{14} (x^2-4x+2) \end{aligned} \quad (1.3.14)$$

$$\begin{aligned} > n:=subs(isolate(SkPr(o[3],n),c),n); \\ n:=x^3 - \frac{6}{5} + \frac{43}{5} x - 6x^2 \end{aligned} \quad (1.3.15)$$


$$\begin{aligned} > n := -6/5+43/5*x-6*x^2+x^3; \\ n:=x^3 - \frac{6}{5} + \frac{43}{5} x - 6x^2 \end{aligned} \quad (1.3.16)$$

$$\begin{aligned} > \text{map}(i \rightarrow \text{SkPr}(n, o[i]), [\$1..3]); \\ \text{map}(i \rightarrow \text{SkPr}(n, x^i), [\$0..2]); \\ & \quad [0, 0, 0] \\ & \quad [0, 0, 0] \end{aligned} \quad (1.3.17)$$

$$\begin{aligned} > \text{SkPr}(n, n); \\ & \quad \frac{72}{5} \end{aligned} \quad (1.3.18)$$


> plot([x^3,n],x=0..4,color=[red,green]);





Man könnte  $\sqrt{\text{SkPr}(n, n)}$  als den **Abstand** von  $x^3$  von dem Teilraum  $\langle 1, x, x^2 \rangle$  bezeichnen. Ebenso könnte man  $x^3 - n$ , also das Bild der Orthogonalprojektion, als eine **Approximation** von  $x^3$  durch Polynome vom Grad  $< 3$  *im Sinne unseres Skalarproduktes* ansprechen:

```
> plot([x^3, x^3-n], x=0..4, color=[red, blue]);
```



**MATH:** Haben wir eine Orthonormalbasis  $B \in V^n$  unseres Euklidischen Vektorraumes  $(V, \Psi)$ , so gilt für alle  $v \in V$ :

$$v = \Psi(v, B_1) \cdot B_1 + \dots + \Psi(v, B_n) \cdot B_n$$

Die  $\Psi(v, B_i)$  heißen auch **Fourier-Koeffizienten** von  $v$ .

**MATH:** Was wir aus der obigen Formel kennenlernen, sind

**Orthogonalprojektionen:** Ist  $W$  ein Teilraum von  $V$  und  $B$  in  $V^n$  eine Orthonormalbasis von  $V$ , so dass  $(B_1, \dots, B_k)$  eine Basis (und damit Orthonormalbasis) von  $W$  ist, so heißt

$$\pi: V \rightarrow V: v \mapsto \Psi(v, B_1) \cdot B_1 + \dots + \Psi(v, B_k) \cdot B_k$$

die **Orthogonalprojektion** von  $V$  auf  $W$ .

Dass Orthogonalprojektionen in der Tat Projektionen sind, können wir leicht nachrechnen:

### ÜBUNG [03]:

Zeige: Die Orthogonalprojektion ist eine Projektion (d.h. ist  $\pi$  eine Orthogonalprojektion, so gilt  $\pi^2 = \pi$ ).

**MATH:** Orthogonalprojektionen sind charakterisiert als die Projektionen, bei denen Kern und Bild orthogonal zueinander sind.

#### ÜBUNG [04]:

Zeige: Die Definition der Orthogonalprojektion ist unabhängig von der gewählten Basis  $B$ .

*Hinweis:* Zeige für eine weitere "passende" ONB  $C$  gilt:  $({}^C Id^B)^{-1} = ({}^C Id^B)^{tr}$  und teile die beteiligten Matrizen in Blöcke auf.

**MATH:** Beachte: Eine Orthogonalprojektion ist bereits durch ihren Bildraum festgelegt.

Um ein Gefühl für Orthogonalprojektionen zu bekommen, schauen wir uns ein Beispiel an, mit dem wir bereits vertraut sind:

#### ÜBUNG [05]:

- 1) (Wiederholung) Gib alle Projektionen von  $\mathbb{R}^3 \times 1$  (durch ihre Matrizen bezüglich der Standardbasis) an, deren Bild von  $(1, 1, 1)^{tr}$  erzeugt wird.
- 2) Welche von diesen ist die Orthogonalprojektion, wenn wir die Standardbasis als Orthonormalbasis nehmen? (Erinnerung: Der Kern der Projektion ist das Komplement des Bildes.)

**DENKANSTOSS:** Warum sind die Matrizen von Orthogonalprojektionen bezüglich ONBasen immer symmetrisch, d. h. gleich ihren Transponierten?

**MATH:** Jedenfalls sieht man auch leicht aus der Formel

$$v = \Psi(v, B_1) \cdot B_1 + \dots + \Psi(v, B_n) \cdot B_n$$

dass das Bild unter der Orthogonalprojektion  $\pi(v)$  das eindeutige Element von  $W = \langle B_1, \dots, B_k \rangle$  ist, welches am nächsten an  $v$  liegt, für das also  $\|v - \pi(v)\|$

$$= \sqrt{\Psi(v - \pi(v), v - \pi(v))} \text{ minimal unter allen Elementen aus } W \text{ ist. Dieser}$$

Wert heißt auch der **Abstand** von  $v$  von  $W$  und  $\pi(v)$  die **beste**

**Approximation** von  $v$  an  $W$ .

Haben wir also erstmal die Orthogonalprojektion von  $V$  auf  $W$  gefunden, so macht sie es uns sehr leicht, die beste Approximation eines Vektors  $v$  an  $W$  auszurechnen:

#### ÜBUNG [06]:

Bestimme (unter Benutzung der Lösung der letzten Aufgabe) die beste

Approximation und den Abstand von  $\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}$  zu dem Raum  $\left\langle \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix} \right\rangle$ , wobei die Standardbasis Orthonormalbasis ist.

## ▼ Fünfeckige Sterne: Anschauliche Anwendung von Eigenwerten

[Aufgaben: 1

> **restart;**  
**with(LinearAlgebra):**  
**with(plots):**

### ▼ Inhalt

> **Z:=SubMatrix(Matrix(5,5)+1,1..5,[2,3,4,5,1]);**

$$Z := \begin{bmatrix} 0 & 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \end{bmatrix} \quad (2.1.1)$$

Leider gibt der einschlägige Befehl von LinearAlgebra die Eigenwerte und Eigenvektoren von Z in einer nicht sonderlich schönen Form aus. Wir bemühen uns daher selbst.

Die Matrix ist die Begleitmatrix von  $x^5 - 1$ . In der Tat:

> **MinimalPolynomial(Z,x);**  
**CharacteristicPolynomial(Z,x);**  
 $x^5 - 1$   
 $x^5 - 1$  (2.1.2)

Wir erkennen also die Eigenwerte als die fünften Einheitswurzeln, also die  $c \in \mathbb{C}$  mit  $c^5 = 1$ . Maple rechnet diese Eigenwerte nicht so elegant aus.

> **convert(Eigenvalues( Z ),list);**  
 $\left[ 1, \frac{1}{4} \sqrt{5} - \frac{1}{4} + \frac{1}{4} I \sqrt{2} \sqrt{5 + \sqrt{5}}, -\frac{1}{4} \sqrt{5} - \frac{1}{4} + \frac{1}{4} I \sqrt{2} \sqrt{5 - \sqrt{5}}, \right.$  (2.1.3)  
 $\left. -\frac{1}{4} \sqrt{5} - \frac{1}{4} - \frac{1}{4} I \sqrt{2} \sqrt{5 - \sqrt{5}}, \frac{1}{4} \sqrt{5} - \frac{1}{4} - \frac{1}{4} I \sqrt{2} \sqrt{5 + \sqrt{5}} \right]$

Hier sind die Eigenvektoren:

```
> V:=map(j->Vector(5,i->exp(2*Pi*I*(i-1)*(-j)/5)),[$0..4]);
```

$$V := \begin{bmatrix} 1 \\ 1 \\ 1 \\ 1 \\ 1 \end{bmatrix}, \begin{bmatrix} 1 \\ e^{-\frac{2}{5}I\pi} \\ e^{-\frac{4}{5}I\pi} \\ e^{\frac{4}{5}I\pi} \\ e^{\frac{2}{5}I\pi} \end{bmatrix}, \begin{bmatrix} 1 \\ e^{-\frac{4}{5}I\pi} \\ e^{\frac{2}{5}I\pi} \\ e^{-\frac{2}{5}I\pi} \\ e^{\frac{4}{5}I\pi} \end{bmatrix}, \begin{bmatrix} 1 \\ e^{\frac{4}{5}I\pi} \\ e^{-\frac{2}{5}I\pi} \\ e^{\frac{2}{5}I\pi} \\ e^{-\frac{4}{5}I\pi} \end{bmatrix}, \begin{bmatrix} 1 \\ e^{\frac{2}{5}I\pi} \\ e^{\frac{4}{5}I\pi} \\ e^{-\frac{4}{5}I\pi} \\ e^{-\frac{2}{5}I\pi} \end{bmatrix} \quad (2.1.4)$$

Probe:

```
> map(j->simplify(Z.V[j+1]-exp(2*Pi*I*j/5).V[j+1]),[$0..4]);
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix} \quad (2.1.5)$$

Wir wollen diese Eigenvektoren graphisch darstellen.

Hier ist ein Programm, welches einen komplexen 5-Vektor  $\in \mathbb{C}^5$  in eine Punktfolge der Länge 5 im  $\mathbb{R}^2$  umwandelt, die man als Input für das Polygonplotprogramm nehmen kann:

```
> vis:=proc(V::Vector)
```

```
>   convert(map(a->[Re(a),Im(a)],V),list);
```

```
> end proc;
```

```
> vis(V[2]);
```

$$\left[ [1, 0], \left[ \cos\left(\frac{2}{5}\pi\right), -\sin\left(\frac{2}{5}\pi\right) \right], \left[ -\cos\left(\frac{1}{5}\pi\right), -\sin\left(\frac{1}{5}\pi\right) \right], \left[ -\cos\left(\frac{1}{5}\pi\right), \sin\left(\frac{1}{5}\pi\right) \right], \left[ \cos\left(\frac{2}{5}\pi\right), \sin\left(\frac{2}{5}\pi\right) \right] \right] \quad (2.1.6)$$



Wir fassen eine Punktfolge der Länge 5 im  $\mathbb{R}^2$  als einen "Stern" auf, dessen Ecken wir ausgehend von dem ersten Punkt in zyklischer Reihenfolge durchlaufen. Anwenden unserer Matrix Z bedeutet etwas für die Reihenfolge der Punkte der Folge. Hingegen die Multiplikation mit einer komplexen Zahl bedeutet die Drehstreckung der Folge um das Zentrum 0. Addition zweier Folgen ist wieder offensichtlich: Entsprechende Punkte werden addiert. Da die Eigenwerte allesamt den Betrag 1 haben, ist die Bedeutung der Eigenfolgen diese: induzierte Umnummerierung hat den selben Effekt wie eine gewisse



Drehung. Die möglichen Winkel entsprechen den Eigenwerten.

Also entspricht ein Vektor  $\in \mathbb{C}^5$  einem Stern, welcher durch die Darstellung in der Eigenvektorbasis in 5 Bausteine, den Sternen entsprechend der Eigenvektoren, zerlegt werden kann.

Unsere vier Hauptbausteine für die Sterne:

```
> display(array(1..2,1..2,  
[  
[  
  polygonplot(vis(V[2]),scaling=CONSTRAINED,color=blue),  
  polygonplot(vis(V[3]),scaling=CONSTRAINED,color=blue)  
] ,  
[  
  polygonplot(vis(V[4]),scaling=CONSTRAINED,color=blue),  
  polygonplot(vis(V[5]),scaling=CONSTRAINED,color=blue)  
]  
])  
));
```





Der erste Baustein, den wir nicht gemalt haben, ist einfach ein fünffach belegter Punkt, den man sich als Schwerpunkt der zu konstruierenden Sterne vorstellen darf. Er kommt aus dem Eigenvektor  $(1, 1, 1, 1, 1)$ .

**MATH:** Der Schwerpunkt  $s$  eines Polygons mit den Eckpunkten  $x_1, \dots, x_n$  ist

definiert als das arithmetische Mittel dieser Ecken:  $s := \frac{1}{n} \cdot \sum_{i=1}^n x_i$ .

Die Inverse zu

> **T:=Matrix(V);**

(2.1.7)

$$T := \begin{bmatrix} 1 & 1 & 1 & 1 & 1 \\ 1 & e^{-\frac{2}{5}i\pi} & e^{-\frac{4}{5}i\pi} & e^{\frac{4}{5}i\pi} & e^{\frac{2}{5}i\pi} \\ 1 & e^{-\frac{4}{5}i\pi} & e^{\frac{2}{5}i\pi} & e^{-\frac{2}{5}i\pi} & e^{\frac{4}{5}i\pi} \\ 1 & e^{\frac{4}{5}i\pi} & e^{-\frac{2}{5}i\pi} & e^{\frac{2}{5}i\pi} & e^{-\frac{4}{5}i\pi} \\ 1 & e^{\frac{2}{5}i\pi} & e^{\frac{4}{5}i\pi} & e^{-\frac{4}{5}i\pi} & e^{-\frac{2}{5}i\pi} \end{bmatrix} \quad (2.1.7)$$

kann man zwar mit Maple ausrechnen. Aber wir wollen lieber eine ästhetisch ansprechendere Form:

> **TI:=1/5\*Matrix(5,5,(i,j)->simplify(conjugate(T[j,i])));**

$$TI := \begin{bmatrix} \frac{1}{5} & \frac{1}{5} & \frac{1}{5} & \frac{1}{5} & \frac{1}{5} \\ \frac{1}{5} & \frac{1}{5} (-1)^{2/5} & \frac{1}{5} (-1)^{4/5} & \frac{1}{5} e^{-\frac{4}{5}i\pi} & \frac{1}{5} e^{-\frac{2}{5}i\pi} \\ \frac{1}{5} & \frac{1}{5} (-1)^{4/5} & \frac{1}{5} e^{-\frac{2}{5}i\pi} & \frac{1}{5} (-1)^{2/5} & \frac{1}{5} e^{-\frac{4}{5}i\pi} \\ \frac{1}{5} & \frac{1}{5} e^{-\frac{4}{5}i\pi} & \frac{1}{5} (-1)^{2/5} & \frac{1}{5} e^{-\frac{2}{5}i\pi} & \frac{1}{5} (-1)^{4/5} \\ \frac{1}{5} & \frac{1}{5} e^{-\frac{2}{5}i\pi} & \frac{1}{5} e^{-\frac{4}{5}i\pi} & \frac{1}{5} (-1)^{4/5} & \frac{1}{5} (-1)^{2/5} \end{bmatrix} \quad (2.1.8)$$

Probe:

> **simplify(TI.T);**

$$\begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix} \quad (2.1.9)$$

Jetzt kann das Spiel beginnen! Wir unterscheiden die Koordinatenspalten bezüglich der Standardbasis, die direkt die Punktfolge angeben, und die Koordinatenspalten bezüglich der Eigenvektorbasis  $V$ , die uns die Zerlegung unseres Sterns (oder allgemeiner Polygons) in unsere 5 Bausteine angeben. Erste Spalten (also die Koordinatenspalten bezüglich der Standardbasis) nennen wir **Punktspalten**, letztere (also die Koordinatenspalten bezüglich der




Eigenvektorbasis  $V$ ) **Spektralspalten**. Also

**$T \cdot I$** . Punktspalte = Spektralspalte

**$T$** . Spektralspalte = Punktspalte.


Die Koeffizienten in der Spektralspalte sind komplexe Zahlen, die eine Drehstreckung mit dem Bausteinern durchführen. Z. B.:

```
> display([  
  polygonplot(vis(expand(V[3])),scaling=CONSTRAINED,color=  
  blue),  
  polygonplot(vis(expand((1+i)*V[3])),thickness=3,scaling=  
  CONSTRAINED,color=red)  
]);
```



Wir wollen jetzt den regelmäßigen Stern  $V[3]$  leicht stören durch  $V[2]$ :

```
> polygonplot(vis(T.Vector(5,[0,1/10*I,1,0,0])),scaling=  
  CONSTRAINED,color=blue);
```



An diesem Stern nehmen wir jetzt von Hand kosmetische Operationen vor:

```
> W:=T.Vector(5,[0,1/10*I,1,0,0]);
> evalf(W);
```


$$\begin{bmatrix} 1. + 0.1000000000I \\ -0.7139113428 - 0.5568835528I \\ 0.3677955190 + 0.8701548171I \\ 0.2502384686 - 1.031958216I \\ -0.9041226460 + 0.6186869516I \end{bmatrix} \quad (2.1.10)$$

Wir ziehen die erste Spitze raus:

```
> W[1]:=3*W[1];
```

$$W_1 := 3 + \frac{3}{10} I \quad (2.1.11)$$

```
> polygonplot(vis(W),scaling=CONSTRAINED,color=blue);
```



Jetzt ist natürlich interessant zu erfahren, wie sich diese Manipulation auf unsere Spektralkoeffizienten ausgewirkt hat:

```
> w:=evalf(TI.W);
```


$$w := \begin{bmatrix} 0.3999999998 + 0.03999999998I \\ 0.4000000003 + 0.1399999999I \\ 1.400000000 + 0.03999999988I \\ 0.3999999998 + 0.0400000001I \\ 0.4000000003 + 0.0400000001I \end{bmatrix} \quad (2.1.12)$$

Wir können wieder den Koordinatenursprung als Schwerpunkt wählen:

```
> w[1]:=0;
```

$$w_1 := 0 \quad (2.1.13)$$

```
> polygonplot(vis(T.w),scaling=CONSTRAINED,color=blue);
```



Wir wollen den Stern weniger barock haben:


```
> w[2]:=0;w[5]:=0.5*w[5];
```

```
      w2 := 0
```

```
      w5 := 0.2000000002 + 0.02000000005 I
```

(2.1.14)

```
> polygonplot(vis(T.w),scaling=CONSTRAINED,color=blue);
```



### ÜBUNG [07]:

- 1) Verstehe die obigen Sterne. Was ist der Zusammenhang zwischen Punktspalten und Spektralspalten? Welche Rechnungen kann man besser in den Spektralspalten und welche besser in den Punktspalten machen?
- 2) Wie entstehen die Sterne aus der Matrix  $\mathbf{Z}$ ? Was macht die Matrix  $\mathbf{Z}$  mit einem 5-Tupel? Warum betrachten wir die Eigenvektoren der Matrix  $\mathbf{Z}$  und warum sind dies (manchmal) Sterne? Welche Bedeutungen haben die Eigenwerte bei den entsprechenden Sternen?
- 3) Wie sind die Koeffizienten der Eigenvektoren zu wählen, damit der resultierende Vektor nicht durch einen Stern, bei dem sich viele Kanten schneiden, representiert wird, sondern durch ein Gebilde, bei dem sich die Kanten nur in den Eckpunkten schneiden.
- 4) Kann man es in Teilaufgabe (3) auch so einrichten, dass der umschriebene Bereich zusätzlich nicht konvex ist, d. h. zwei Punkte hat, deren Verbindungsstrecke nicht vollständig innerhalb dieses Bereiches verläuft?

└ └ └ **PROJEKT:** Was ist neu, wenn wir  $n$ -eckige Sterne betrachten, wo  $n$  keine Primzahl mehr ist.